BIT manuscript No.
(will be inserted by the editor)

Exact BDF Stability Angles with Maple

Martin J. Gander · Gerhard Wanner

Received: date / Accepted: date

(Dedicated to the 75th anniversary of Syvert P. Nørsett)

Abstract BDF formulas are among the most efficient methods for numerical integration, in particular of stiff equations (see e.g. Gear [2]). Their excellent stability properties are known for precisely half a century, from the first calculation of their angles of $A(\alpha)$-stability by Nørsett [4]. Later, more insight was gained and more precise values were calculated numerically (see for example [3, Sect. V.2]). This was the state-of-the-art, when Akrivis and Katsoprinakis [1] discovered *exact* values for these angles. In this note we simplify the derivation and results by using Maple.

Keywords Stiff differential equations · $A(\alpha)$-stability · BDF methods

Mathematics Subject Classification (2000) 65L07

Let $z = x + iy = e^{i\varphi}$ lie on the unit circle, then the *root locus curve* for k-step BDF is

$$w = u + iv = \sum_{j=1}^{k} \frac{1}{j}(1 - z^{-1})^j\,, \qquad (0 \le \varphi \le 2\pi)$$

(see e.g., [3, Sect. V.1, Eq. (1.17)]). It describes the boundary of the stability domain (see Figure 1). We use the classical parametrization of the unit circle

$$x = \frac{1-t^2}{1+t^2}\,, \quad y = \frac{2t}{1+t^2}\,, \quad t = \tan\frac{\varphi}{2}\,, \qquad (-\infty < t < +\infty)$$

Martin J. Gander
Section of Mathematics, University of Geneva,
E-mail: martin.gander@unige.ch

Gerhard Wanner
Section of Mathematics, University of Geneva,
E-mail: gerhard.wanner@unige.ch



Fig. 1 Root locus curve and stability domain for BDF4.

(see e.g., Euler's *Calculus Integralis*, Caput V, § 261, 1768). The angle α of $A(\alpha)$-stability is then computed by the following Maple commands:

```
k:=3;                                          # possible values 3,4,5,6
x:=(1-t^2)/(1+t^2); y:=2*t/(1+t^2);            # circle parametrization
w:=simplify(sum(1/j*(1-x+I*y)^j,j=1..k));      # root locus curve w
u:=simplify(evalc(Re(w)));                     # real part of w
v:=simplify(evalc(Im(w)));                     # imaginary part of w
p:=simplify(v/(-u));                           # tangent of alpha
pd:=simplify(diff(p,t));                       # derivative of p
pdn:=factor(numer(pd));                        # factor the numerator
t0:=solve(op(2,pdn),t);                        # find zeros
p0:=subs(t=t0[1],p);                           # for k=5 need t0[3]
alpha:=evalf(arctan(p0)/Pi*180);               # evaluate angle
```

Here, $p = \tan\alpha = -\frac{v}{u}$, and pdn is the numerator of $\frac{dp}{dt}$, a polynomial in t^2 due to symmetry, whose largest root t_0^2 we have to compute (corresponding to the extremal value w_0 furthest away from the origin). For $k = 3, 4$ and 6, this t_0^2 is rational, which leads to expressions for $p_0 = \tan\alpha$ containing only one square root (with `Digits:=30`):

k	t_0^2	$p_0 = \tan\alpha$	α
3	$\frac{9}{35}$	$\frac{329}{135}\sqrt{35}$	86.032366860211647332387423479°
4	$\frac{2}{3}$	$\frac{699}{512}\sqrt{6}$	73.351670474578482110409536864°
5	$\times$	$\times$	51.839755836049910391602721533°
6	$\frac{15}{13}$	$\frac{45503}{1974375}\sqrt{195}$	17.839777792245700101632480553°

The case $k = 5$ is more complicated, since here[1]

$$pdn = 15(248t^4 - 275t^2 + 25)(t^2+1)^4, \quad \text{largest root:} \quad t_0^2 = \frac{275}{496} + \frac{5}{496}\sqrt{2033}\,.$$

[1] The factor $(t^2+1)^4$ is related to the order of the method; a similar factor appears for all k and is the deeper reason why the computation of t_0 is so easy.

Therefore we have to simplify the expression for p_0 by using the field structure of the set $\{a+b\sqrt{2033}\ ;\ a,b \text{ rationals}\}$. The command `evala(simplify(p0))` in Maple leads to

$$p_0 = \left(-\frac{51844971}{14086400} + \frac{5765167}{70432000}\sqrt{2033}\right)\sqrt{8525+155\sqrt{2033}}.$$

Remark. By differentiating x (instead of p), we obtain exact values for the values of D in Gear's definition of *stiff stability* (see [2] or [3, p.250]) as

k	3	4	5	6
D	$\frac{1}{12}$	$\frac{2}{3}$	$\frac{93}{80}+\frac{25}{48}\sqrt{5}$	$\frac{243}{40}$

.

References

1. G. Akrivis and E. Katsoprinakis, *Maximum Angles of $A(\vartheta)$-stability of Backward Difference Formulae*, BIT (2019) DOI: 10.1007/s10543-019-00768-1.
2. C.W. Gear, *Numerical initial value problems in ordinary differential equations*, Prentice Hall 1971.
3. E. Hairer, G. Wanner, *Solving Ordinary Differential Equations, Vol. II*, Springer, 2nd ed. 1996.
4. S.P. Nørsett, *A Criterion for $A(\alpha)$-stability of Linear Multistep Methods*, BIT 9 (1969), 259–263.